


Stesura a nome di: Andrea Stornello e Alessandro Adonia


Riprendiamo la lezione di giorno 02/03/2018 e consideriamo la struttura composta da 2 corpi rigidi collegati da due travi, nello specifico due profilati channel (detti a "c").

Poniamo al centro della struttura un sistema di orientamento globale (O, x, y, z) e applichiamo ai due corpi due coppie torcenti C e C' , è possibile riconoscere la forma di struttura simmetrica e il carico asimmetrico;




utilizzando i piani $y-z$ di anti-simmetria per il carico e $x-z$ di simmetria per la struttura, è possibile studiare la struttura prendendo in analisi un quarto di essa compresa tra i punti $OABD$.

Chiaramente i punti compresi tra B e D' sono tutti appartenenti a piani di sezione ortogonale alla trave, infatti rappresentando in vista frontale la sezione al punto D del concio di trave si riesce ad identificare sia la terna di assi locali centrata nel baricentro $G(,,)$ che il centro di taglio C scostato di una quantità e in direzione ξ .



Si riesce così a rappresentare, con le dovute considerazioni il quarto di struttura realmente studiato (in cui, specifichiamo, AB sono collegati rigidamente, BD sono collegati dalla trave e l'asse tratteggiato è il luogo dei centri di taglio)



Le sollecitazioni applicate alla struttura dall'esterno risulteranno: la coppia $\frac{C}{2}$ pari a metà della coppia totale in quanto l'altra metà verrà scaricata sull'immagine di struttura non rappresentata e le reazioni vincolari (legati alla continuità del materiale) generate dall'interazione tra questa e le altre porzioni di struttura;

queste ultime sono delle forze entropiano introdotte per evitare moti di traslazione, nello specifico:

- W_A forza in direzione z applicata in A
- U_A forza in direzione x applicata in A
- θ_A momento di reazione vincolare in direzione y applicato in A
- W_D forza in direzione z applicata in D
- V_D forza in direzione x applicata in D
- θ_D momento di reazione vincolare in direzione x applicato in D

Passando ora alla stesura su maxima per inserire le equazioni di equilibrio alle traslazioni rispetto ad x,y,z indicizziamo alla voce eqtx, eqty, eqtz i risultati delle equazioni corrispondenti

(Comando "kill(all);" usato per pulire la memoria

comando ":" utilizzato per richiamare successivamente un'equazione

tramite i comandi veloci: " Shift+Invio " eseguiamo la singola cella e "Ctrl+R" eseguiamo tutte le celle)

```
[ --> kill( all ) ;
  (%00) done
```

```
[ --> eqtz : WA + WD = 0 ;
  (%01) WD+WA=0
```

```
[ --> eqtx : UA = 0;
  (%02) UA=0
```

```
[ --> eqty : VD = 0;
  (%03) VD=0
```

le equilibri alle rotazioni (relativi ai moti di corpo rigido) passanti per polo O

```
--> eqr0x : + C/2      + UA*0      + WA*0
              + ThetaD      + VD*0      + WD*(b+e)
              = 0;
(%o4) (e+b)WD + C/2 + ThetaD = 0
(%i5) eqr0y : PhiA - a * WA ;
(%o5) PhiA - a WA
(%i6) eqr0z : 0 = 0;
(%o6) 0 = 0
```

Creo quindi un sistema di equazioni usando:

- una lista di equazioni che chiamo "eqns"
- una lista di incognite "unks"

(intesa dal programma come una serie di sotto-espressioni separate da virgole e comprese tra parentesi quadre)

```
(%i7) eqns : [ eqtx , eqty , eqtz , eqr0x , eqr0y , eqr0z ] ;
(%o7) [UA=0, VD=0, WD+WA=0, (e+b)WD + C/2 + ThetaD=0, PhiA-a WA, 0=0]
(%i8) unks : [ UA , WA , PhiA , VD, WD, ThetaD ] ;
(%o8) [UA, WA, PhiA, VD, WD, ThetaD]
```

E tramite il comando "linsolve" tento di risolvere "eqns" equazioni in "unks" incognite;

```
--> linsolve ( eqns , unks ) ;
solve: dependent equations eliminated: (6)
(%o9) [UA=0, WA=-%r1, PhiA=-%r1 a, VD=0, WD=%r1, ThetaD = -C+2 %r1 e+2 %r1 b / 2]
```

eseguendo notiamo che sistema presenta ∞^1 soluzioni (cerchiata in rosso) in quanto l'equazione dipendente (eqr0z) è stata eliminata riducendo il sistema a 5 equazioni e 6 incognite.

Creo allora una lista ridotta di incognite in cui non inserisco "ThetaD" che tratto come parametro ed eseguo nuovamente, associando al termine "sols" il risultato del "linsolve"

```
--> unks : [ UA , WA , PhiA , VD, WD] ;
(%o10) [UA, WA, PhiA, VD, WD]
--> sols : linsolve ( eqns, unks );
solve: dependent equations eliminated: (6)
(%o11) [UA=0, WA = C+2 ThetaD / 2 e+2 b, PhiA = a C+2 a ThetaD / 2 e+2 b, VD=0, WD = -C+2 ThetaD / 2 e+2 b]
--> sols : linsolve ( eqns, unks );
solve: dependent equations eliminated: (6)
(%o11) [UA=0, WA = C+2 ThetaD / 2 e+2 b, PhiA = a C+2 a ThetaD / 2 e+2 b, VD=0, WD = -C+2 ThetaD / 2 e+2 b]
```

In questo modo troviamo una forza WA, effettuo una veloce analisi dimensionale per verificare che il

procedimento sia corretto $\left[\frac{\frac{N}{mm} + \frac{N}{mm}}{mm+mm} \right]$

Ora posso precede in 2 modi:

opzione 1:

```
WA : ev ( WA , sols );
UA : ev ( UA , sols );
PhiA : ev ( PhiA , sols );
VD : ev ( VD , sols );
WD : ev ( WD , sols );
```

valutando con il comando “ev” (valuta) i termine tenendo conto sia del sistema di sostituzioni globali che le sostituzioni indicizzate dal “sols”, determinando così ad uno ad uno ciascuno dei simboli si arriva alla soluzione secondo il sistema di equazioni risolto;

```
(%i12) WA : ev ( WA , sols );
(%o12)  $\frac{C+2\,ThetaD}{2\,e+2\,b}$ 

(%i13) UA : ev ( UA , sols );
(%o13) 0

(%i14) PhiA : ev ( PhiA , sols );
(%o14)  $\frac{a\,C+2\,a\,ThetaD}{2\,e+2\,b}$ 

(%i15) VD : ev ( VD , sols );
(%o15) 0

(%i16) WD : ev ( WD , sols );
(%o16)  $-\frac{C+2\,ThetaD}{2\,e+2\,b}$ 
```

questa procedura risulta però laboriosa e complessa quando il sistema meccanico richiede la determinazione di diversi parametri, è quindi più comodo implementare un sistema automatico per la determinazione dei parametri.

opzione 2

risolvo le equazioni con opzione aggiuntiva

"globalsolve=True"

risolvendo le equazioni con un'azione aggiuntiva mai vista prima, il comando “globalsolve” cioè una variabile che il codice del “linsolve” testa in esecuzione se:

- globalsolve=False (default), non assegna alle incognite il valore soluzione del sistema
- globalsolve=True assegno automaticamente alle incognite il valore soluzione per il sistema

in particolare nel valutare l'espressione “linsolve” si tiene conto dell'ipotesi dettata dal “globalsolve” (un'identità ulteriore) se questa non è specificata sarà applicata come falso logico di default, se questa viene indicata si tiene in conto (è richiesto però di eseguire dall'inizio l'intero processo con Ctrl+R)

```
(%i12) linsolve ( eqns, unks ) , globalsolve=True ;
solve: dependent equations eliminated: (6)
(%o12) [UA:0,WA: $\frac{C+2\,ThetaD}{2\,e+2\,b}$ ,PhiA: $\frac{a\,C+2\,a\,ThetaD}{2\,e+2\,b}$ ,VD:0,WD: $-\frac{C+2\,ThetaD}{2\,e+2\,b}$ ]
```

notiamo infatti che “UA:0” cioè UA non solo uguale a zero ma a zero è stata assegnata un’identità.

Verifico poi che l’assegnazione sia avvenuta correttamente

$$\begin{cases} \text{(\%i13)} \quad W_A; \\ \text{(\%o13)} \quad \frac{C+2 \, \text{Theta}D}{2 \, e+2 \, b} \end{cases}$$


$$\begin{cases} \text{(\%i14)} \quad W_D; \\ \text{(\%o14)} \quad -\frac{C+2 \, \text{Theta}D}{2 \, e+2 \, b} \end{cases}$$

In questo modo ho valori noti per tutte le reazioni vincolari per il carico esterno.


Ora bisogna valutare quel “ThetaD” che permette di ottenere una quantità di rotazione in D nulla, si dovrà quindi scrivere l’equazione di rotazione al punto D uguale a zero (per continuità) e più nello specifico per il quarto di struttura sarà una rotazione al punto D (in direzione x) uguale a zero (equazione all’antisimmetria); per far questo si utilizza Castigliano che, come ricordiamo, richiede l’energia potenziale elastica della struttura, quindi le caratteristiche delle sollecitazioni della struttura (cioè momento flettente e taglio, momento torcente e sforzo normale).

Passiamo quindi alle caratteristiche di sollecitazione; nel tratto AB corpo rigido trovo energia potenziale elastica nulla per cui non si calcolano momento flettente, momento torcente e taglio, mentre nel concio di trave compreso nel tratto BD effettuo lo studio dei tre parametri.


- $[N = 0]$ Lo sforzo normale è nullo in quanto la sezione è assialmente scarica
- $[T_\eta = -W_D]$ Il taglio secondo l’asse locale η risulta essere uguale, in valore assoluto, alla reazione vincolare $|W_d|$ ma essendo una forza entropiano della faccia opposta al sistema di riferimento locale si assume in valore negativo




- $[T_\xi = -V_D]$ Il taglio secondo l’asse locale ξ viene calcolato in modo analogo al precedente come mostrato in figura




- $[M_\xi = -W_D \cdot s]$ Per il momento flettente M_ξ associato a ξ si trova come unico carico rilevante W_D (in quanto θ_D è un momento torcente per il tratto BD); si procede allo studio sezionando il concio di trave di lunghezza s e applicando l'equilibrio risulta necessario l'impiego di W_D a distanza s cioè una coppia ($W_D \cdot s$), essendo questa coppia rivolta in direzione opposta al sistema di riferimento dell'asse locale ξ (come visto precedentemente) verrà presa negativa




- $[M_\eta = V_D \cdot s]$ Analogamente il momento flettente M_η associato a η si ottiene da una forza V_D posta a distanza s .



- $[M_t = \theta_D]$ Infine per il momento torcente della trave trovo che tra i carichi rilevanti ci sono i due tagli che essendo passanti dal centro di taglio hanno braccio nullo e una coppia θ_D applicata in D, con la procedura di equilibrio al troncamento della sezione "s" quindi bisogna applicare un momento θ_D in direzione assiale opposta alla faccia positiva, risulta così in segno negativo.



Ricapitolando $\begin{cases} N = 0 \\ T_{\eta} = -W_D \\ T_{\xi} = -V_D \end{cases}$ e $\begin{cases} M_{\xi} = -W_D \cdot s \\ M_{\eta} = +V_D \cdot s \\ M_t = \theta_D \end{cases}$



Inserisco ora le caratteristiche di sollecitazione

```
--> N : 0;
(%015) 0

--> T_eta : -WD ;
(%016) (C+2 ThetaD) / (2e+2b)

--> T_xi : -VD ;
(%017) 0
```

N.B. Solitamente si scrive lo sforzo normale per ogni singolo tratto di trave, in questo caso dovrei scrivere N_{BD} ma essendo l'unico risulta superfluo sottolinearlo

Passo quindi ad analizzare momenti bisogna ricorrere alla funzione di "ascissa curvilinea" che in questo caso è una quota rettilinea e scorre lungo la trave, con $0 \leq s \leq a$ considerata adimensionale.

Momenti flettenti e torcente

```
--> Mt : -ThetaD;
(%018) -ThetaD

--> Mf_xi : -WD*s;
(%019) s(C+2 ThetaD) / (2e+2b)

--> Mf_eta : +VD*s;
(%020) 0
```

Si passa quindi al calcolo dell'energia potenziale elastica.

Posto che:

- la sezione in oggetto ha assi ξ , η principali d'inerzia in quanto simmetrica

- il materiale supposto omogeneo e isotropo

Per calcolare la densità lineica dell'energia potenziale associata a: $M_\xi, M_\eta, M_t, N, T_\xi, T_\eta$

$$dU_{ds} = \frac{M_\xi^2}{2 \cdot E \cdot J_{\xi\xi}} + \frac{M_\eta^2}{2 \cdot E \cdot J_{\eta\eta}} + \frac{M_t^2}{2 \cdot G \cdot K_t} + \frac{N^2}{2 \cdot E \cdot Area} + \frac{(\chi_\xi \cdot T_\xi)^2}{2 \cdot G \cdot Area} + \frac{(\chi_\eta \cdot T_\eta)^2}{2 \cdot G \cdot Area}$$

Con: χ_ξ, χ_η coefficienti correttivi; $J_{\xi\xi}, J_{\eta\eta}, K_t$ moduli di rigidezza torsionale della sezione; E modulo di Jang, G modulo di taglio.

Una volta ottenute le espressioni delle caratteristiche di sollecitazione su tutti i tratti di trave, procedo alla valutazione dell'energia potenziale elastica sulla struttura (facendo la sommatoria di tutti i contributi energetici delle porzioni).

densità lineica di en. pot. elastica

```
--> dU_ds :      Mf_xi ^2 / 2 / E / J_xixi
                +      Mf_eta^2 / 2 / E / J_etaeta
                +      Mt   ^2 / 2 / G / Kt
                +      N    ^2 / 2 / E / area
                + chi_xi * T_xi ^2 / 2 / G / area
                + chi_eta * T_eta ^2 / 2 / G / area ;
(%o21) chi_eta(C+2 ThetaD)^2 ThetaD^2 + s^2 (C+2 ThetaD)^2
        2area(2e+2b)^2 G      2KtG      2(2e+2b)^2 J_xixiE
```

Ed eseguendo (Shift+Invio) troverò come incognite $\chi_\xi, Area, K_t, J_{\xi\xi}$

N.B. Se la trave è molto snella per cui il taglio ha un'influenza secondaria rispetto al momento flettente e al momento torcente, in termini di contenuti energetici, ci permette di trascurare il calcolo dei due χ , questo avviene anche quando il taglio rapportato alle proprietà della sezione e del materiale (come nei non isotropi cedevoli al taglio)

Una volta nota la densità lineica dell'energia potenziale elastica su ogni tratto di trave basta integrarla sulla trave di lunghezza a e si ottiene l'energia potenziale totale. Definendo con il termine "U" il risultato dell'integrale calcolato con la funzione "integrate(, , ,)" in cui: il primo termine è l'argomento dell'integrale tra parentesi, il secondo termine è la variabile di integrazione, il terzo e il quarto corrispondono agli estremi integrazione

integro la densità lineica di en. pot. el. e ottengo l'en. pot. elastica della struttura.

```
(%i22) U : integrate ( dU_ds , s , 0 , a );
(%o22) ((a^3 area Kt C^2 + 4 a^3 area Kt ThetaD C + 4 a^3 area Kt ThetaD^2) G + (3 a chi_eta J_xixi Kt C^2
+ 12 a chi_eta J_xixi Kt ThetaD C +
(12 a chi_eta J_xixi Kt + (12 a area e^2 + 24 a area b e + 12 a area b^2) J_xixi) ThetaD^2) E) / (
(24 area e^2 + 48 area b e + 24 area b^2) J_xixi Kt E G)
```

E' possibile applicare ora il teorema di Castigliano (per cui lo spostamento in direzione della forza del punto di applicazione è uguale alla derivata parziale della forza agente rispetto all'energia elastica di deformazione)

nota l'espressione per l'energia pot. el. della struttura, applico il teorema di Castigliano per ricavare la componente x della rotazione della sezione in D

Con ThetaD coppia applicata alla sezione studiata concorde in direzione e verso della direzione che si vuole monitorare.

Derivando quindi l'energia potenziale elastica (U) nella coppia applicata "ThetaD" (Θ_d) e si ottiene la rotazione "thetaD" (θ_d), chiamata appunto in modo diverso tra loro.

Applicando per la prima volta la funzione di derivazione spieghiamo che viene intesa dal programma con il comando "diff(f(x) ,x ,n)" in cui il primo argomento è l'espressione da derivare, il secondo argomento è la variabile di derivazione ed in fine il terzo argomento per indicare l'ordine di derivazione; in più per derivate miste basta posizionare la funzione a più variabili da derivare sempre al primo termine per poi inserire a 2 a 2 variabile e ordine di derivazione esempio: "diff(f(xy) ,x ,n, y, m);".

```
(%i23) thetaD : diff( U , ThetaD );
(%o23) ((4 a^3 area Kt C + 8 a^3 area Kt ThetaD) G + (12 a chi_eta J_xixi Kt C + 2
(12 a chi_eta J_xixi Kt + (12 a area e^2 + 24 a area b e + 12 a area b^2) J_xixi) ThetaD) E) / (
(24 area e^2 + 48 area b e + 24 area b^2) J_xixi Kt E G)
```

Il risultato verrà trovato in funzione dei parametri di carico, dei moduli elastici e delle proprietà di sezione viste precedentemente

Ponendo in fine la rotazione $\theta_d=0$

ricavo quel valore di ThetaD che annulla la rotazione thetaD ripristinando la continuità del materiale al piano di antisimmetria yz tra la porzione di struttura modellata e la sua immagine

```
(%i24) eq_thetaD : thetaD = 0;
(%o24) ((4 a^3 area Kt C + 8 a^3 area Kt ThetaD) G + (12 a chi_eta J_xixi Kt C + 2
(12 a chi_eta J_xixi Kt + (12 a area e^2 + 24 a area b e + 12 a area b^2) J_xixi) ThetaD) E) / (
(24 area e^2 + 48 area b e + 24 area b^2) J_xixi Kt E G) = 0
```

Effettuando poi una derivazione del secondo ordine nell'incognita "ThetaD" (Θ_d) ci si assicura della linearità della stessa e procedendo con "linsolve(risolvi l'equazione, nell'incognita), con globalsolve true;" si trova finalmente il valore cercato.

```
(%i25) diff(eq_thetaD , ThetaD, 2);
(%o25) 0=0

(%i26) linsolve( eq_thetaD , ThetaD ) , globalsolve=True;
(%o26) [ThetaD:-
a^2 area Kt C G + 3 chi_eta J_xixi Kt C E
-----]
2 a^2 area Kt G + (6 chi_eta J_xixi Kt + (6 area e^2 + 12 area b e + 6 area b^2) J_xixi) E
```

Il "thetaD" (θ_d) viene trovato in funzione della coppia C, il cui valore si determina applicando nuovamente il teorema di Castigliano. Andando a controllare l'energia potenziale elastica si vede come questa sia ancora funzione sia di ThetaD che di C (anche se ormai ThetaD stesso è funzione di C).

Basta chiedere quindi al programma un'ulteriore valutazione di U in funzione solo di C utilizzando il comando "fullratsimp()", che permette di effettuare semplificazioni svolgendo la somma a denominatore comune.

forzo la sostituzione dei termini ThetaD presenti entro U secondo la sua più recente definizione con la forma

ThetaD = ThetaD(C)

appena trovata

```
(%i27) U :  
      fullratsimp(  
          ev( U )  
      );
```

```
(%o27) 
$$\frac{a^3 \text{area} C^2 G + 3 a \text{chi\_eta} J_{xixi} C^2 E}{8 a^2 \text{area} K t G^2 + (24 \text{chi\_eta} J_{xixi} K t + (24 \text{area} e^2 + 48 \text{area} b e + 24 \text{area} b^2) J_{xixi}) E G}$$

```


Avendo ridefinito precedentemente U come “U: fullratsimp(ev());” l’energia potenziale elastica della struttura completa, che chiamo “Ustruttcomp”, sarà quindi pari a 4 volte l’energia potenziale elastica di un quarto di struttura

```
(%i28) Ustruttcomp : 4 * U;
```

```
(%o28) 
$$\frac{4 (a^3 \text{area} C^2 G + 3 a \text{chi\_eta} J_{xixi} C^2 E)}{8 a^2 \text{area} K t G^2 + (24 \text{chi\_eta} J_{xixi} K t + (24 \text{area} e^2 + 48 \text{area} b e + 24 \text{area} b^2) J_{xixi}) E G}$$

```

Applicando quindi alla struttura completa una coppia ed una contro-coppia e sapendo che il lavoro compiuto è dato dall’entità delle forze per la rotazione relativa fra i due corpi rigidi si è trovato il valore dell’energia potenziale elastica della struttura completa;



per semplificare lo studio si può vincolare uno dei due corpi rigidi e considerare una rotazione totale della struttura intesa non più come rotazione relativa bensì come rotazione assoluta. E’ necessario però a questo punto precisare che la rigidità torsionale da prendere in considerazione non sarà più K_T riferita alla sola trave ma risulterà pari a $K_{\text{Struct}} = \frac{C}{\theta}$.

La rotazione dell'intera struttura è determinata derivando l'energia potenziale elastica della struttura completa per la coppia (sempre tramite il teorema di Castigliano)

```
(%i29) theta_strutt : diff( Ustruttcomp , C );
```

$$(\%o29) \frac{4 \left(2 a^3 \text{area} C G + 6 a \chi_{\eta} J_{xixi} C E \right)}{8 a^2 \text{area} K_t G^2 + \left(24 \chi_{\eta} J_{xixi} K_t + \left(24 \text{area} e^2 + 48 \text{area} b e + 24 \text{area} b^2 \right) J_{xixi} \right) E G}$$

La rigidezza torsionale, che chiamo "Kt_strutt", sarà data dal rapporto tra coppia torsionale applicata e rotazione torsionale indotta

```
(%i30) Kt_strutt : fullratsimp(
    C/theta_strutt
);
```

$$(\%o30) \frac{a^2 \text{area} K_t G^2 + \left(3 \chi_{\eta} J_{xixi} K_t + \left(3 \text{area} e^2 + 6 \text{area} b e + 3 \text{area} b^2 \right) J_{xixi} \right) E G}{a^3 \text{area} G + 3 a \chi_{\eta} J_{xixi} E}$$

è stato inserito anche il comando di semplificazione "fullratsimp" perché se la risposta è lineare la rigidezza torsionale non è funzione del carico applicato. Risulta inoltre che la rigidezza alla torsione della struttura è funzione delle quote geometriche macroscopiche (a,b,e), della rigidezza a flessione $J_{\xi\xi}$, della rigidezza a torsione della trave K_t e della rigidezza a taglio tenendo conto del χ_{ξ} .

Per tirar fuori uno dei valori specifici si dovrebbero sostituire gli stessi valori calcolati per una qualche sezione con cui costruiamo il corpo, ad esempio prendendo qualche estremo, provando ad esempio quando la rigidezza torsionale della struttura si annulla,


```
(%i31) limit ( Kt_strutt, Kt , 0 );
```

$$(\%o31) \frac{\left(3 \text{area} e^2 + 6 \text{area} b e + 3 \text{area} b^2 \right) J_{xixi} E G}{a^3 \text{area} G + 3 a \chi_{\eta} J_{xixi} E}$$

ma il risultato non è ancora nullo .

Il fatto che le travi siano 2 fa sì che la loro rigidezza flessionale a taglio possa supplire alla mancanza della loro rigidezza torsionale, quindi nella struttura potremmo valutare oltre che l'impiego di profili a sezione chiusa (o per meglio dire in parete sottile chiusa) che hanno alta rigidezza torsionale, anche all'impiego di profili a parete sottile aperta, che hanno bassissima rigidezza torsionale, in quanto si può sfruttare la loro rigidezza flessionale a taglio per avere una risposta rigida (come ad esempio i profili a doppia T).

Riprendendo quindi il caso di semplificazione in cui uno dei due corpi è vincolato e al secondo viene applicata una coppia C




per ognuna delle due travi è possibile tirare fuori dei parametri quali:

1. Rigidezza a torsione pari a GK_T
2. Rigidezza a taglio pari a $GA \frac{1}{\chi}$
3. Rigidezza a flessione pari a $EJ_{\xi\xi}$

che possono essere studiate e schematizzate tramite le rappresentazioni serie-parallelo per sottolinearne la dipendenza relativa l'una dall'altra anche se non è sempre facile determinare i particolari legami quindi non è sempre possibile ridurre strutture complesse a forme di rigidezza torsionale serie parallelo.

Analizzando ad esempio una struttura del tipo rappresentata in figura basta che una delle due rigidezze flessionale o taglia vada a zero per annullare l'influenza dell'altra




Analizzando tutto questo con wxMaxima e portando a zero sia il taglio che il momento flettente con un semplice $\lim_{K_T \rightarrow 0} K_{Strutt}$

```
(%i31) limit ( Kt_strutt, Kt , 0 );
(%o31) 
$$\frac{(3 \text{area} e^2 + 6 \text{area} b e + 3 \text{area} b^2) J_{xixi} E G}{a^3 \text{area} G + 3 a \chi_{\eta} J_{xixi} E}$$

```


Si ottiene un valore diverso da zero che, teoricamente, riesce a reggere da solo la struttura.

Ricordiamo che la resistenza della rigidezza a torsione è data dall'annullamento degli spostamenti verticali come mostrato in figura




Per cui la deformazione della trave avverrà seguendo:


- Il principio di deformazione puramente flessionale in cui lo spostamento è dato anche sotto l'ipotesi di forza nulla se $EJ_{\xi\xi} = 0$ e riscontriamo un punto di flesso centrato nella trave



- Il principio di deformazione puramente tagliante in cui si ottiene sempre lo stesso spostamento elementare ottenuto sotto l'ipotesi di forza nulla se $GA \frac{1}{\chi} = 0$



Risulta quindi necessario analizzare la sezione di trave per impostare il foglio di calcolo per i 3 segmenti (anima e ali)



Determinate le lunghezze b e h possiamo dire che il sistema è simmetrico, per semplificare i calcoli vengono poi piazzati 2 sistemi di riferimento

- OXY centrato nella mezzeria dell'anima e utile per determinare le coordinate nodali
- Gxy centrato sul baricentro (che in questo caso giace sull'asse di simmetria) impiegato per svolgere più semplicemente i calcoli

Risulta quindi possibile ottenere:

- J_{xx} (inteso fino ora come $J_{\xi\xi}$), J_{xy} e J_{yy}
- L'area

- Il K_T
- Il χ_y (inteso fin ora come χ_ξ) χ_x e χ_{xy} nel caso di sezione simmetrica

Sul foglio di calcolo si inizierà considerando le lunghezze suddette b e h più la sezione di spessore costante t

```
[ (%i10) kill(all);
  (%o0) done
```

```
[ dichiaro le quote dimensionali essere ristrette in segno, in modo da avere |x| = x
```

```
[ considero per semplicità uno spessore costante t
```

```
[ (%i1) assume(b>0,t>0,h>0);
  (%o1) [b>0, t>0, h>0]
```

Definiti i 4 punti come coordinate rispetto al sistema OXY

```
[ piazzo 2 sistemi
  OXY centrato nella mezzeria dell'anima
  Gxy centrato nel baricentro
```

```
[ inserendo coordinate dei punti nodali del profilo
```

```
[ (%i2) XA : b $ YA: h/2$
        XB : 0 $ YB: h/2$
        XC : 0 $ YC:-h/2$
        XD : b $ YD:-h/2$
```

Si possono determinare le lunghezze dei 3 tratti della sezione con Pitagora

```
[ derivo lunghezze dei tratti
```

```
[ (%i10) l_AB : sqrt((XA-XB)^2 + (YA-YB)^2)$
         l_BC : sqrt((XB-XC)^2 + (YB-YC)^2)$
         l_CD : sqrt((XC-XD)^2 + (YC-YD)^2)$
```

L'implementazione del codice richiede di poter effettuare studio delle caratteristiche proprie della sezione, del materiale e della reazione alle forze esterne sui vari tratti; risulta necessario allora poter sezionare uno di essi in qualsiasi punto tramite l'utilizzo dell'ascissa curvilinea "s".

Ad esempio per il tratto AB posto che l'ascissa curvilinea inizi da A verso B per identificare univocamente le procedure a disposizione sono 2.

1. Definire una derivata curvilinea S con valore pari a zero in A e crescente di una lunghezza
 - b per arrivare al punto B
 - b+h per arrivare al punto C
 - 2b+h per arrivare al punto D
2. Definire una coordinata ξ per ogni tratto (con $0 < \xi < 1$), risulta necessario definire le lunghezze dei tratti ξ in S, per trasformare ξ in S si pone che:
 - Per il tratto AB un $S = l_{AB} \cdot \xi$
 - Per il tratto BC un $S = l_{AB} + l_{BC} \cdot \xi$
 - Per il tratto CD un $S = l_{AB} + l_{BC} + l_{CD} \cdot \xi$

(così che integrando per ogni tratto dS sarà uguale al $d\xi$ per la lunghezza del tratto stesso)

Passando al foglio di calcolo tutte le precedenti considerazioni per associare ad ogni punto interno al tratto una funzione, prese delle coordinate in funzione di x e y, e usando il comando “:=” si assegna la funzione ad un’espressione che ha bisogno, per essere risolta, di essere traslata in fine in coordinate baricentriche.

(ad esempio per il tratto AB preso un punto qualunque identificato univocamente da delle coordinate che saranno funzione di xi in modo tale da rispettare: $x_{AB}(0) = x_A$ e $x_{AB}(1) = x_B$)

```
[ associo delle coordinate, funzione di un'ascissa curvilinea adimensionale xi in [0,1]
(%i13) x_AB(xi) := (XB-XG)*xi + (XA-XG)*(1-xi)$
      y_AB(xi) := (YB-YG)*xi + (YA-YG)*(1-xi)$
      x_BC(xi) := (XC-XG)*xi + (XB-XG)*(1-xi)$
      y_BC(xi) := (YC-YG)*xi + (YB-YG)*(1-xi)$
      x_CD(xi) := (XD-XG)*xi + (XC-XG)*(1-xi)$
      y_CD(xi) := (YD-YG)*xi + (YC-YG)*(1-xi)$
```

Verificando e valutando in mezzeria le coordinate si può usare il comando “ev” con “fullratsimp” per ridurre a forma numeratore su denominatore o “expand” per ridurre a forma di monomi

```
[ (%i25) fullratsimp(ev(x_AB(xi) , xi=1/2 ));
(%o25)  $-\frac{2XG-b}{2}$ 
(%i26) expand(ev(x_AB(xi) , xi=1/2 ));
(%o26)  $\frac{b}{2}-XG$ 
```

O più semplicemente sostituire direttamente alla ξ un valore d’interesse come

```
[ (%i28) expand(x_AB(1/2));
(%o28)  $\frac{b}{2}-XG$ 
```

Conoscendo tutte le coordinate rispetto al sistema baricentrico si possono calcolare le aree per integrazione $a_{AB} = \int_0^1 t l_{AB} d\xi$

```
[ calcolo aree per integrazione sulla sezione
(%i19) area : integrate(t * l_AB ,xi,0,1)
          + integrate(t * l_BC ,xi,0,1)
          + integrate(t * l_CD ,xi,0,1);
(%o19)  $h t + 2 b t$ 
```

momento statico del primo ordine

```
(%i20) Mx : integrate(y_AB(xi) * t * l_AB ,xi,0,1)
          + integrate(y_BC(xi) * t * l_BC ,xi,0,1)
          + integrate(y_CD(xi) * t * l_CD ,xi,0,1);

      My : integrate(x_AB(xi) * t * l_AB ,xi,0,1)
          + integrate(x_BC(xi) * t * l_BC ,xi,0,1)
          + integrate(x_CD(xi) * t * l_CD ,xi,0,1);

(%o20) 
$$\frac{b t (2 YG + h)}{2} - \frac{b t (2 YG - h)}{2} - h t YG$$

(%o21) 
$$-b t (2 XG - b) - h t XG$$

```

derivo baricentro imponendo momento statico nullo
se calcolato su sistema di riferimento baricentrico

```
(%i22) linsolve([Mx=0,My=0],[XG,YG]), globalsolve=true;

(%o22) [XG:  $\frac{b^2}{h+2b}$ , YG: 0]
```